4. PARAMETRYCZNE TESTY ISTOTNOŚCI

Lp.	*Hipoteza H_0*	*Założenia*	*Statystyka testowa*	*Rozkład statystyki testowej*
1.	$\mu=\mu_0$	znane σ lub nieznane σ i duża próba	$U=\frac{\bar{x}-\mu}{\sigma}\sqrt{n}$	$\mathcal{N}(0, 1)$ dla nieznanego σ: $\sigma\approx S$ lub $\sigma\approx s$
2.	$\mu=\mu_0$	nieznane σ i mała próba	$t=\frac{\bar{x}-\mu}{s}\sqrt{n}$	t – *Studenta* o $(n-1)$ stopniach swobody
3.	$\mu_1=\mu_2$	znane σ_1 i σ_2	$U=\frac{\bar{x}_1-\bar{x}_2}{\sqrt{\frac{\sigma_1^2}{n_1}+\frac{\sigma_2^2}{n_2}}}$	$\mathcal{N}(0, 1)$
4.	$\mu_1=\mu_2$	nieznane ale równe σ_1 i σ_2	$t=\frac{\bar{x}_1-\bar{x}_2}{s\sqrt{\frac{1}{n_1}+\frac{1}{n_2}}}$, *gdzie:* $s=\sqrt{\frac{(n_1-1)s_1^2+(n_2-1)s_2^2}{(n_1-1)+(n_2-1)}}$	rozkład t – *Studenta* o (n_1-n_2-2) stopniach swobody
5.	$\sigma_1^2=\sigma_2^2$		$F=\frac{s_1^2}{s_2^2}$, $(s_1^2>s_2^2)$	rozkład F o $v_1=n_1-1$ i $v_2=n_2-1$ stopniach swobody
6.	$p=p_0$	duża próba $(n\geq 100)$	$U=\frac{\hat{p}-p}{\sqrt{p(1-p)/n}}$.	$\mathcal{N}(0, 1)$
7.	$p_1=p_2$	duża próba $(n_1,n_2\geq 100)$	$U=\frac{p_1-p_2}{\sqrt{\bar{p}(1-\bar{p})/n}}$, *gdzie:* $p_1=m_1/n_1$, $p_2=m_2/n_2$, $\bar{p}=(m_1+m_2)/(n_1+n_2)$, $n=n_1n_2/(n_1+n_2)$	$\mathcal{N}(0, 1)$

4.1 Testy parametryczne w Excel-u

Test dla μ (znane σ)

TEST.Z(liczby; śred; odch)

gdzie:

liczby – dane na podstawie których weryfikowana jest hipoteza

H_0: **μ** = **śred** wobec hipotezy alternatywnej

H_1: **μ** > **śred**;

śred – testowana wartość średniej; **odch** – odchylenie standardowe rozkładu.

Funkcja **TEST.Z** zwraca graniczny poziom istotności *p–value* dla testu prawostronnego, dla testu lewostronnego należy wyliczyć wartość:

1-TEST.Z(liczby; śred; odch)

dla testu dwustronnego:

2*MIN(TEST.Z(liczby; śred; odch); 1-TEST.Z(liczby; śred; odch))

Test o równości średnich $\mu_1 = \mu_2$ (nieznane ale równe σ_1 i σ_2)

TEST.T(liczby1; liczby2; ślady; typ)

gdzie:

liczby1, liczby2 – dane na podstawie których weryfikowana jest hipoteza,

postać hipotezy zależy od parametru **ślady**, jeśli **ślady** ustalono na 1:

testowana jest hipoteza H_0: $\mu_1 \geq \mu_2$ wobec hipotezy alternatywnej H_1: $\mu_1 < \mu_2$,

(lub H_0: $\mu_1 \leq \mu_2$ H_1: $\mu_1 > \mu_2$),

jeśli **ślady** ustalono na 2:

testowana jest hipoteza H_0: $\mu_1 = \mu_2$ wobec hipotezy alternatywnej H_1: $\mu_1 \neq \mu_2$.

jeśli **typ=1** przeprowadzany jest test sparowany; jeśli **typ=2** przyjmowane jest założenie że odchylenia standardowe w obydwu populacjach są równe, jeśli **typ=3** zakłada się że odchylenia są różne.

Funkcja **TEST.T** zwraca graniczny poziom istotności *p–value* .

Test o równości wariancji $\sigma_1^2 = \sigma_2^2$

TEST.F(liczby1; liczby2)

gdzie:

liczby1, liczby2 – dane na podstawie których weryfikowana jest hipoteza,

Funkcja testuje hipotezę H_0: $\sigma_1^2 = \sigma_2^2$ wobec hipotezy alternatywnej H_1: $\sigma_1^2 \neq \sigma_2^2$.

Funkcja **TEST.F** zwraca graniczny poziom istotności *p–value* dla testu dwustronnego, wartość *p–value* dla testu jednostronnego otrzymuje się po dwukrotnym zmniejszeniu wartość otrzymanej dla testu dwustronnego.

Uwaga! Pozostałe testy istotności przeprowadza się korzystając z wzorów (patrz punkt 4.1.), część testów dostępna jest także z poziomu dostępnego w Excel-u dodatku ***Analysis ToolPak***.

Dodatek Analysis ToolPak

Weryfikację hipotezy o równości średnich dwóch populacji można również wykonać wykorzystując dostępną w Excelu funkcję „**Analiza danych**”. Funkcja ta jest dostępna z menu **Narzędzia|Analiza danych**.

Uwaga! Analiza danych domyślnie nie jest w Excelu włączona. Żeby włączyć funkcję należy wybrać w menu opcję **Narzędzia|Dodatki** i w wyświetlanym oknie zaznaczyć pole wyboru: **Analysis ToolPak**.



Analysis ToolPak dostarcza wielu różnych narzędzi, dostępne są między innymi niektóre testy istotności, np.:

- test o równości średnich $\mu_1 = \mu_2$ w przypadku gdy znane są σ_1 i σ_2,
- test o równości wariancji $\sigma_1^2 = \sigma_2^2$.



Przykład 1.

Rozkład pomiarów długości detalu jest rozkładem normalnym z odchyleniem standardowym $\sigma = 1.5$. . Wykonano 10 niezależnych pomiarów długości pewnego detalu otrzymując wyniki: 18, 21, 22,4, 23, 21,3, 21,9, 17,6, 21, 17,8 i19,4. Zweryfikować na poziomie istotności $\alpha = 0.01$ hipotezę, że rzeczywista długość detalu wynosi 20.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n**	10									
3	**poziom istotności α**	0,01									
4	**odchylenie standardowe σ**	1,5									
5	**weryfikowana średnia**	20									
6											
7	**x_{sr}**	**=ŚREDNIA(B1:K1)**									
8	**un**	**=(B7-B5)/B4*PIERWIASTEK(B2)**					**un=0,71678**				
9											
10	**obszar krytyczny**										
11	**ua**	**=-ROZKŁAD.NORMALNY.S.ODW(B3/2)**						**ua=2,57583**			
12											
13	**p-value**										
14	**1/2 p-value**	**=1-ROZKŁAD.NORMALNY.S(B8)**						**=TEST.Z(B1:K1;B5;B4)**			
15	**p-value**	**=2*B14**					**p-value=0,47351**				

Przykład 2.

Automat produkuje detale o nominalnej długości 20. Wykonano 10 niezależnych pomiarów długości pewnego detalu otrzymując wyniki: 20, 21, 22.4, 23, 21.3, 21.9, 20.6, 21, 19.8, 20.4. Czy obliczona średnia długość detalu równa 21.14 pozwala na stwierdzenie, że rzeczywista długość jest większa od nominalnej. Przyjąć poziom istotności $\alpha = 0.01$.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary**	20	21	22,4	23	21,3	21,9	20,6	21	19,8	20,4
2	**Rozmiar próby n**	10									
3	**poziom istotności α**	0,01									
4	**weryfikowana średnia**	20									
5											
6	**x_{sr}**	**=ŚREDNIA(B1:K1)**									
7	**odchylenie standardowe σ**	**=ODCH.STANDARD.POPUL(B1:K1)**									
8	**tn**	**=(B6-B4)/B7*PIERWIASTEK(B2-1)**					**tn=3,48616**				
9											
10	**obszar krytyczny**										
11	**ta**	**=ROZKŁAD.T.ODW(B3*2;B2-1)**					**ta=2,82143**				
12											
13	**p-value**	**=ROZKŁAD.T(B8;B2-1;1)**					**p-value=0,00344**				

Przykład 3.

Wykonano dwie serie pomiarów długości detalu z jednakową dokładnością $\sigma_1 = \sigma_2 = 1.5$. W pierwszej serii przeprowadzono 10 pomiarów: 18, 21, 22.4, 23, 21.3, 21.9, 17.6, 21, 17.8, 19.4 a w drugiej 8 pomiarów: 22.1, 20.3, 21.4, 23.1, 21.1, 21.8, 20.6, 22.8. Zweryfikować na poziomie istotności $\alpha = 0.01$ hipotezę, że rozbieżność średnich jest nieprzypadkowa.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary1**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n1**	10									
3	**odchylenie standardowe σ1**	1,5									
4											
5	**Pomiary2**	22,1	20,3	21,4	23,1	21,1	21,8	20,6	22,8		
6	**Rozmiar próby n2**	8									
7	**odchylenie standardowe σ2**	1,5									
8											
9	**poziom istotności α**	0,01									
10											
11	x_{sr1}	**=ŚREDNIA(B1:K1)**									
12	x_{sr2}	**=ŚREDNIA(B5:I5)**								**un=-1,84115**	
13	**un**	**=(B11-B12)/PIERWIASTEK(POTĘGA(B3;2)/B2+POTĘGA(B7;2)/B6)**									
14											
15	**obszar krytyczny**										
16	**ua**	**=-ROZKŁAD.NORMALNY.S.ODW(B9/2)**						**ua=2,57583**			
17											
18	**p-value**										
19	**1/2 p-value**	**=ROZKŁAD.NORMALNY.S(B13)**									
20	**p-value**	**=2*B19**					**p-value=0,0656**				

Weryfikację hipotezy o równości średnich dwóch populacji w przypadku gdy znane są odchylenia standardowe można przeprowadzić również korzystając z ***dodateku Analysis ToolPak*** po wybraniu opcji: *Test z: z dwiema próbami dla średnich.*



W wyświetlonym oknie należy:

- wskazać obszar z danymi pierwszej populacji,
- wskazać obszar z danymi drugiej populacji,
- w przypadku badania hipotezy zerowej o równości średnich pole „Różnica średnich wg hipotezy” wypełnić wartością 0,
- określić wariancje dla pierwszej i drugiej populacji,
- określić graniczny poziom istotności,
- wskazać „opcje wyjścia” czyli miejsce w którym zwrócone zostaną obliczenia wykonane podczas testowania hipotezy.



Poniższy arkusz zawiera otrzymane wyniki obliczeń.

	A	B	C	D	E	F	G	H
21	Test z: z dwiema próbami dla średnich							
22								
23		*Zmienna 1*	*Zmienna 2*					
24	Średnia	20,34	21,65					
25	Znana wariancja	2,25	2,25					
26	Obserwacje	10	8					
27	Różnica średnich wg hipotezy	0						
28	z	-1,841148327						
29	P(Z<=z) jednostronny	0,032799913						
30	Test z jednostronny	2,326347874						
31	P(Z<=z) dwustronny	0,065599825						
32	Test z dwustronny	2,575829304						

Przykład 4.

Rozwiązać zadanie z przykładu 3. zakładając, że odchylenia standardowe populacji σ_1 i σ_2 nie są znane.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary1**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n1**	10									
3	**Pomiary2**	22,1	20,3	21,4	23,1	21,1	21,8	20,6	22,8		
4	**Rozmiar próby n2**	8									
5	**poziom istotności α**	0,01									
6											
7	x_{sr1}	**=ŚREDNIA(B1:K1)**									
8	x_{sr2}	**=ŚREDNIA(B3:I3)**									
9	s_1	**=ODCH.STANDARDOWE(B1:K1)**									
10	s_2	**=ODCH.STANDARDOWE(B3:I3)**									
11	**n1+n2-2**	**=B2+B4-2**									
12	**s**	**=PIERWIASTEK(((B2-1)*POTĘGA(B9;2)+(B4-1)*POTĘGA(B10;2))/B11)**									
13	**tn**	**=(B7-B8)/(B12*PIERWIASTEK(1/B2+1/B4))**						→	**tn= -1,68849**		
14	**obszar krytyczny**										
15	**ta**	**=ROZKŁAD.T.ODW(B5;B11)**					→	**ta= 2,92078**			
16											
17	**p-value**										
18	**1/2 p-value**	**=ROZKŁAD.T(-B13;B11;1)**									
19	**p-value**	**=2*B18**				→	**p-value= 0,11071**				
20	**p-value**	**=ROZKŁAD.T(-B13;B11;2)**									
21	**p-value**	**=TEST.T(B1:K1;B3:I3;2;2)**									

Weryfikację hipotezy o równości średnich dwóch populacji w przypadku gdy nieznanych ale równych odchyleń standardowych można przeprowadzić również korzystając z ***dodateku Analysis ToolPak*** po wybraniu opcji: *Test t: z dwiema próbami zakładający równe wariancje.*



W wyświetlonym oknie należy:

- wskazać obszary z danymi pierwszej i drugiej populacji,
- w przypadku badania hipotezy zerowej o równości średnich pole „Różnica średnich wg hipotezy" wypełnić wartością 0,
- określić graniczny poziom istotności,

- wskazać „opcje wyjścia” czyli miejsce w którym zwrócone zostaną obliczenia wykonane podczas testowania hipotezy.



Otrzymane wyniki obliczeń zostały zwrócone w postaci arkusza:.

	A	B	C	D	E	F	G	H
25	Test t: z dwiema próbami zakładający równe wariancje							
26								
27		*Zmienna 1*	*Zmienna 2*					
28	Średnia	20,34	21,65					
29	Wariancja	3,984888889	0,99142857					
30	Obserwacje	10	8					
31	Wariancja sumaryczna	2,67525						
32	Różnica średnich wg hipotezy	0						
33	df	16						
34	t Stat	-1,688487501						
35	P(T<=t) jednostronny	0,055354337						
36	Test T jednostronny	2,583487179						
37	P(T<=t) dwustronny	0,110708675						
38	Test t dwustronny	2,920781621						

Przykład 5.

Wykonano dwie serie pomiarów długości detalu uzyskując wyniki takiej jak w przykładzie 3. Na poziomie istotności $\alpha = 0.01$ zweryfikować hipotezę o jednakowej wariancji obydwu serii pomiarów.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary1**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n1**	10									
3	**Pomiary2**	22,1	20,3	21,4	23,1	21,1	21,8	20,6	22,8		
4	**Rozmiar próby n2**	8									
5	**poziom istotności α**	0,01									
6											
7	s_1	**=ODCH.STANDARDOWE(B1:K1)**									
8	s_2	**=ODCH.STANDARDOWE(B3:I3)**									
9	**Fn**	**=POTĘGA(B7;2)/POTĘGA(B8;2)**				→	**Fn = 4,01934**				
11											
12	**obszar krytyczny**										
13	**Fa**	**=ROZKŁAD.F.ODW(B5;B2-1;B4-1)**				→	**Fa = 6,718752**				
14											
15	**p-value**										
16	**p-value**	**=ROZKŁAD.F(B9;B2-1;B4-1)**				→	**p-value= 0,040077**				
17	**p-value**	**=TEST.F(B1:K1;B3:I3)/2**									

Do testowania hipotezy o równości wariancji można wykorzystać również **Analysis ToolPak**. W tym przypadku należy skorzystać z testu: *Test t: z dwiema próbami dla wariancji.*



Uzyskane wyniki są zwracane w postaci arkusza:

	A	B	C	D	E	F	G	H
18	Test F: z dwiema próbami dla wariancji							
19								
20		*Zmienna 1*	*Zmienna 2*					
21	Średnia	20,34	21,65					
22	Wariancja	3,984889	0,991429					
23	Obserwacje	10	8					
24	df	9	7					
25	F	4,01934						
26	P(F<=f) jednostronny	0,040077						
27	Test F jednostronny	6,718752						

Przykład 6.

Wysunięto hipotezę, że wadliwość pewnego podzespołu wynosi 10%. W celu sprawdzenia tej hipotezy wylosowano próbkę 100 podzespołów i otrzymano w niej 15 podzespołów wadliwych. Zweryfikować hipotezę na poziomie istotności $\alpha = 0.05$.

	A	B	C	D	E	F	G	H	I	J	K
1	**Rozmiar próby n**	100									
2	**Ilość wadliwych**	15									
3	**poziom istotności α**	0,05									
4	**weryfikowana frakcja**	0,1									
5											
6	**frakcja**	**=B2/B1**									
7	**un**	**=(B6-B4)/PIERWIASTEK(B4*(1-B4)/B1)**							**un=1,66667**		
8											
9	**obszar krytyczny**										
10	**ua**	**=-ROZKŁAD.NORMALNY.S.ODW(B3/2)**						**ua=1,95996**			
11											
12	**p-value**										
13	**1/2 p-value**	**=1-ROZKŁAD.NORMALNY.S(B7)**									
14	**p-value**	**=2*B14**						**p-value=0,09558**			

Przykład 7.

Wysunięto hipotezę, że jakość produkcji pewnego wyrobu po wprowadzeniu nowej technologii nie uległa zmianie. Wylosowano 120 sztuk wyprodukowanych starą technologią i otrzymano 12 sztuk wadliwych, wśród wylosowanych 160 sztuk wyprodukowanych nową technologią i otrzymano 20 sztuk wadliwych. Zweryfikować na poziomie istotności α = 0.05 hipotezę o jednakowym wskaźników braków przy produkcji obu metodami.

	A	B	C	D	E	F	G	H	I	J	K
1	**Rozmiar próby 1**	120									
2	**Ilość wadliwych**	12									
3	**Rozmiar próby 2**	160									
4	**Ilość wadliwych**	20									
5	**poziom istotności α**	0,05									
6	**p1**	**=B2/B1**									
7	**p2**	**=B4/B3**									
8	**$p_{śr}$**	**=(B2+B4)/(B1+B3)**									
9	**n**	**=B1*B3/(B1+B3)**									
10	**un**	**=(B6-B7)/PIERWIASTEK(B8*(1-B8)/B9)**									
11											
12	**obszar krytyczny**										
13	**ua**	**=ROZKŁAD.NORMALNY.S.ODW(B5/2)**									
14											
15	**1/2 p-value**	**=ROZKŁAD.NORMALNY.S(B10)**									
16	**p-value**	**=2*B15**									